Biofuel markets

Ferike Thom

Thünen Institute of Farm Economics, Braunschweig

These lecture slides are based on the teaching material by Mihaly Himics for the same course in 2022.

Content

1. Background
2. Biofuels in CAPRI
    1. Bioethanol and Biodiesel
    2. Supply
    3. Demand

    } Concept, GAMS implementation, mix of biofuel inputs, mix of bio- and fossil fuels
3. Exercise: Simulate a change in the EU biofuel policy

1. Background

- Renewable alternative to fossil fuels, but…
- Sustainability considerations:
  - Energy efficiency (scale of operations)
  - Indirect Land Use Change (ILUC) causing food price increase and loss in carbon sinks
  - Biodiversity loss from deforestation
- EU policy: Renewable Energy Directive (RED II) targets for renewable energy
  - 32% by 2030 overall
  - 14% in transport sector
  - Max. 7% share for biofuels from food crops in transport
  - Increase share of advanced biofuels
  - Sustainability criteria (e.g. certification for low ILUC-risk)

1. Background: Controversies

- Brazil: sugar cane based production (cost efficient); flexible fuel vehicle fleet, sugar-ethanol industry
- US: huge subsidies in the past for 1st generation biofuels (corn-based)
- Policy-driven demand with distorted markets: “Ethanol tourism” between Brazil and USA



Yano et al., 2012

1. Background: Controversies

- Food Security: Did biofuel expansion play a role in 2007/08 food spikes?

The importance of expanded production of biofuels.[8] Estimates of the contribution of increased demand for biofuels, in particular ethanol in the USA,[9] to the price spike vary from a negligible 3% to an overwhelming 75%. The former estimate was derived by estimating the effect on maize prices, in which

1. Background: Controversies

- In Europe: Biodiesel is relatively more important
- Cost efficiency issues relative to the US and Brazil
- Induced deforestation
- palm-oil based biodiesel



Content

1. Background
2. Biofuels in CAPRI
    1. Bioethanol and Biodiesel
    2. Supply
    3. Demand

    } Concept, GAMS implementation, mix of biofuel inputs, mix of bio- and fossil fuels
3. Exercise: Simulate a change in the EU biofuel policy

2. Biofuels in CAPRI

- Bioethanol and biodiesel
- Endogenously modelled demand for biofuels in the transportation sector
- Processing: 1st generation biofuels linked to optimal feedstock mix
- By-products:
  - Distilled Dried Grains with Solubles (DDGS) – animal feed
  - Glycerine – chemical industry

2.1 The ethanol market in CAPRI



Becker, 2011

2.1 The biodiesel market in CAPRI



Becker, 2011

Content

1. Background
2. Biofuels in CAPRI
    1. Bioethanol and Biodiesel
    2. Supply
    3. Demand

    } Concept, GAMS implementation, mix of biofuel inputs, mix of bio- and fossil fuels
3. Exercise: Simulate a change in the EU biofuel policy

2.2 Biofuel production in CAPRI

- Concept on first generation biofuel production
  Biofuel price vs. average cost of biofuel feedstock drives supply



Biofuel supply function in France

Blanco et al., 2013

2.2 Biofuel production in CAPRI

- Synthetic function with three parts



Biofuel supply function in France

linear sigmoid semi-log

Blanco et al., 2013

2.2 Biofuel production in CAPRI

- Synthetic function with three parts: linear, sigmoid and semi-log

gams/arm/market_model.gms l. 1180

```
SupplyBiof_(RMS,XXBIOF) $     ( DATA(RMS,"BiofPriceRel",XXBIOF,"CUR")
                            $ (SUM(Stock_to_Fuel(XBioStock,XXBioF),
                                  DATA(RMS,"BIOF",XBioStock,"CUR")*DATA(RMS,"PRCB",XBioStock,"CUR")))
                            $ ((v_prodBiof.lo(RMS,XXBioF) ne v_prodBiof.up(RMS,XXBioF)) or p_trim)
                            $ (v_prodQuant.lo(RMS,XXBioF) ne v_prodQuant.up(RMS,XXBioF))
                            $ DATA(RMS,"PROD",XXBIOF,"CUR")
                            $ XXX1(XXBIOF)
                            $ p_endoBioMarket
                             ) ..

    v_prodBiof(RMS,XXBioF)/(DATA(RMS,"Prod",XXBioF,"CUR")+0.1)
     =E=
*             --- small linear term to avoid zero production and zero slope..
           (p_bioSupPar(RMS,XXBIOF,"scale")*v_biofPriceRel(RMS,XXBIOF)
*           --- double log part
           + exp(p_bioSupPar(RMS,XXBIOF,"alfa") + p_bioSupPar(RMS,XXBIOF,"beta")*log(v_biofPriceRel(RMS,XXBIOF)+1e-2))
*                  --- multipied to sigmoid function that guarantees steeper slope
                          * SIGMOID(p_bioSupPar(RMS,XXBIOF,"SLOPE")
                                 *(v_biofPriceRel(RMS,XXBIOF)
                                  -p_bioSupPar (RMS,XXBIOF,"Turn"))))/(DATA(RMS,"Prod",XXBioF,"CUR")+0.1);
```

2.2 Biofuel production in CAPRI

- Total biofuel production = sum of
  - first generation,
  - second generation (SECG),
  - non-agricultural (NAGR),
  - and agricultural not covered explicitly (EXOG)

gams/arm/market_model.gms l. 1208

```
MaprBiof_(RMS,XXBioF) $ ((SUM(SAMEAS(XXX,XXBIOF),1)
                        $ (v_prodQuant.lo(RMS,XXBioF) ne v_prodQuant.up(RMS,XXBioF)))
                        $ DATA(RMS,"Prod",XXBioF,"CUR")
                        $ XXX1(XXBIOF) )..

    v_prodQuant(RMS,XXBioF)/(DATA(RMS,"Prod",XXBioF,"CUR")+0.1)
      =E=                       (     v_prodBiof(RMS,XXBioF)
                                  +   DATA(RMS,"NAGR",XXBioF,"CUR")
                                  +   DATA(RMS,"SECG",XXBioF,"CUR")
                                  +   DATA(RMS,"EXOG",XXBioF,"CUR"))/(DATA(RMS,"Prod",XXBioF,"CUR")+0.1);
```

2.2 Biofuel production in CAPRI

- Only 1st generation is endogenous and linked to feedstock in CAPRI (crops modelled endogenously)

gams/arm/market_model.gms l. 1208

```
MaprBiof_(RMS,XXBioF) $ ((SUM(SAMEAS(XXX,XXBIOF),1)
                        $ (v_prodQuant.lo(RMS,XXBioF) ne v_prodQuant.up(RMS,XXBioF)))
                        $ DATA(RMS,"Prod",XXBioF,"CUR")
                        $ XXX1(XXBIOF) )..

    v_prodQuant(RMS,XXBioF)/(DATA(RMS,"Prod",XXBioF,"CUR")+0.1)
      =E=                       (     v_prodBiof(RMS,XXBioF)
                                  +   DATA(RMS,"NAGR",XXBioF,"CUR")
                                  +   DATA(RMS,"SECG",XXBioF,"CUR")
                                  +   DATA(RMS,"EXOG",XXBioF,"CUR"))/(DATA(RMS,"Prod",XXBioF,"CUR")+0.1);
```

2.2 Biofuel production in CAPRI

- Processing of individual feedstocks and optimal mix

STOCK_TO_FUEL(ROWS, *)

	BIOE	BIOD
SWHE	Y	
RYEM	Y	
BARL	Y	
OATS	Y	
MAIZ	Y	
OCER	Y	
TWIN	Y	
WHEA	Y	
SUGA	Y	
RAPO		Y
SUNO		Y
SOYO		Y
PLMO		Y
OTHO		Y

Biofuel Feedstock (crops)
Imported raw material included

2.2 Biofuel production in CAPRI

- Processing of individual feedstocks and optimal mix

gams/arm/market_model.gms l. 1368

```
      v_biofProcQuant(RMS,XX)/(DATA(RMS,"Biof",XX,"CUR")+0.1)

       =E= [
            SUM(Stock_to_fuel(XX,XXBiof),
*
*                ---- a share of output production quantity
*
                 p_dpCESBiof(RMS,XX) * v_prodBiof(RMS,XXBioF)
*
*               ---- times output prices divided by feedcost cost
*                    exponent the substiution elasticity
*
               * (v_biofFeedCost(RMS,XXBIOF) / v_biofFeedCost(RMS,XX)) ** p_rhoBioFuel(RMS,XXBIOF)
             )
           ]/(DATA(RMS,"Biof",XX,"CUR")+0.1);
```

Biofuel average (xxbiof) versus individual feedstock (xx)

2.2 Biofuel production in CAPRI

- Biofuel feedstock balance
- Sum of all feedstock processing = (price-driven) biofuel production

gams/arm/market_model.gms l. 1225

```
prodBiof_(RMS,XXBioF) $ (
                  (SUM(Stock_to_fuel(YbioStock,XXBioF), DATA(RMS,"BioF",YbioStock,"CUR")) GT 1)
*                 --- the numerarie is not fixed
                  $ (NOT SUM(BIOF_NUM(RMS,XXBioF,XX) $ (v_BiofProcQuant.LO(RMS,XX) EQ v_biofProcQuant.UP(RMS,XX)),1))

*                 --- at least one feedstock in data base
                  $  SUM(Stock_to_fuel(XXX,XXBioF) $ (DATA(RMS,"PRCB",XXX,"CUR") $ DATA(RMS,"BioF",XXX,"CUR")),1)
                  $ DATA(RMS,"Prod",XXBioF,"CUR")) ..

    v_prodBiof(RMS,XXBioF)/(DATA(RMS,"Prod",XXBioF,"CUR")+0.1) =E=

        SUM(Stock_to_fuel(XX,XXBioF) $ (DATA(RMS,"PRCB",XX,"CUR") $ DATA(RMS,"BioF",XX,"CUR")),
                                    v_biofProcQuant(RMS,XX) * DATA(RMS,"PRCB",XX,"CUR"))
                          /(DATA(RMS,"Prod",XXBioF,"CUR")+0.1);
*
```

Processing coefficient

2.2 Biofuel production in CAPRI

- Biofuel cost index (average)

gams/arm/market_model.gms l. 1279

```
v_biofFeedCost(RMS,XXBIOF)
  /(DATA(RMS,"BiofFeedCost",XXBIOF,"CUR")+1)
    =E=

          SUM(Stock_to_fuel(XBioStock,XXBiof) $ p_dpCesBiof(RMS,XBioStock),
                p_dpCesBiof(RMS,XBioStock)   *
                (v_biofFeedCost(RMS,XBioStock)/(data(RMS,"BiofFeedCost",XXBIOF,"CUR")+1))
                ** (1-p_rhoBioFuel(RMS,XXBiof))
             ) ** (1/(1-p_rhoBioFuel(RMS,XXBiof)));
```

Content

1. Background
2. Biofuels in CAPRI
    1. Bioethanol and Biodiesel
    2. Supply
    3. Demand

    } Concept, GAMS implementation, mix of biofuel inputs, mix of bio- and fossil fuels
3. Exercise: Simulate a change in the EU biofuel policy

2.3 Biofuel demand in CAPRI

- Share of biofuel in total fuel use. Sigmoid function.
- Total fuel demand exogenous



Biofuel demand share in France

Blanco et al., 2013

2.3 Biofuel demand in CAPRI

fuelMatch(XXBiof, FUEL_ROWS): mapping of biefuels to fossil subsitute		
BIOE	GASL	Y
BIOD	DISL	Y

gams/arm/market_model.gms l. 1391

```
biofDemShare_(RMS,XXBIOF) $ (
                                    SUM(fuelMatch(XXBioF,Fuel_Rows), Data(RMS,"CPRI",fuel_rows,"CUR"))
                                  $ p_bioDemPar(RMS,XXBioF,"Slope")
                                  $ sum(sameas(xxx,xxBiof),1)
                                  $ (v_biofProcQuant.range(RMS,XXBIOF) ne 0)
                                  $ ((v_consShareBioF.range(RMS,XXBIOF) NE 0) or p_trim)
                                  $ xxx1(XXBIOF)
                                  $ DATA(RMS,"BIOF",XXBIOF,"CUR")
                                  $ p_endoBioMarket)..

    v_consShareBioF(RMS,XXBioF)  =E=
*
*       --- fixed share (= blending quota relative)
*
        Data(RMS,"QUTS",XXBIOF,"CUR")
*
*       -- sigmoid function, assumes that rather small shares are used even at high prices,
*          "TURN" reflects the quota
*
      + (SIGMOID(p_bioDemPar(RMS,XXBioF,"Slope")
                          *(v_consprice(RMS,XXBIOF) / SUM(fuelMatch(XXBioF,Fuel_Rows), Data(RMS,"CPRI",fuel_rows,"CUR"))
                           -p_bioDemPar(RMS,XXBioF,"Turn"))))*p_bioDemPar(RMS,XXBioF,"Max");
```

mandate

relative price

2.3 Biofuel demand in CAPRI

- Total use in the country = consumption + feed use + processing + biofuel processing
- Biofuel processing is handled separately from other processing industries
- other industrial use of ethanol is booked under general processing (v_procQuant)

2.3 Biofuel demand in CAPRI

- Total use in the country = consumption + feed use + processing + biofuel processing

gams/arm/market_model.gms l. 2596

```
ArmBal1_(RM,XXX) $ (v_arm1Quant.lo(RM,XXX) ne v_arm1Quant.up(RM,XXX)) ..
*
*    --- total domestic consumption for a trade block
*
$ifi not %mrkExactAgg%==ON  v_arm1Quant(RM,XXX)
$ifi     %mrkExactAgg%==ON  (v_domSales(RM,XXX) + sum(RM1 $ ((not sameas(RM1,RM)) and p_tradeFlows(RM,RM1,XXX,"CUR")), v_tradeFlows(RM,RM1,XXX)))
                            / (DATA(RM,"arm1",XXX,"CUR") $ (not p_trim) + 1 $ p_trim + 0.1)

          =E=  SUM(RMS_TO_RM(RMS,RM),
*                   --- human consumption
                        v_consQuant(RMS,XXX)        $ DATA(RMS,"HCon",XXX,"CUR")
*                   --- feed use
                      + v_feedQuant(RMS,XXX)        $ DATA(RMS,"Feed",XXX,"CUR")
*                   --- explicit processing demand (cakes)
                      + v_procQuant(RMS,XXX)        $ DATA(RMS,"Proc",XXX,"CUR")
*                   --- demand for biofuel processing
                      + v_biofProcQuant(RMS,XXX)    $ (DATA(RMS,"Biof",XXX,"CUR")
*
*                                                     --- in case of sguar market quotas, C-sugar is supposed to
*                                                         be used for biofuels, and has its own market clearing
                                                      $ ( (not sameas(XXX,"SUGA")) or (data(RMS,"QUTS","SUGA","CUR") le eps)))

                  ) / (  DATA(RM,"Arm1",XXX,"CUR")    $ (not p_trim)
                                                    + 1 $ p_trim + 0.1);
*
```

Other industrial use

Biofuels in transportation

Content

1. Background
2. Biofuels in CAPRI
    1. Bioethanol and Biodiesel
    2. Supply
    3. Demand

    } Concept, GAMS implementation, mix of biofuel inputs, mix of bio- and fossil fuels

3. Exercise: Simulate a change in the EU biofuel policy

3. Exercise

- Increase mandate in order to ‘push-up’ total biofuel demand

```
Data(RMS,"QUTS",XXBIOF,"CUR")
```



Biofuel demand share in France

Blanco et al., 2013

3. Exercise: Run the reference file





3. Exercise: Scenario

scenario file calls policy file

Gams/pol_input/
userScens/
mandate_SA_150.gms

```
*******************************************************************************
$ontext

   CAPRI project

   GAMS file : mandate_SA_100.gms

   @purpose  : calculating a demand curve for biofuel processing of crops
               with respect to changing biofuel mandates

   @author   : Mihaly Himics
   @date     : 14.10.19
   @since    :
   @refDoc   :
   @seeAlso  :
   @calledBy :

$offtext
*******************************************************************************

*
*   ---     reference scenario
*

$include pol_input/cap_after_2023/ref.gms

* -------------------------------------------------------------
* --- Fixing pesticide applicatin quantities per ha to circumvent calibration problems
* -------------------------------------------------------------
   DATA(RU,mCact,pestoAgg,'AbsoluteChange") = eps;

*
*   ---     apply a shock on biofuel mandates
*           First argument: multiplier of the original QUTS for biofuels
*

$batinclude 'scen/bio_fuels/change_biof_mandate.gms' 2.5

*  abort 'check scenario shock', data;

```

3. Exercise: Scenario

Policy file takes shock size from scenario file

Gams/scen/biofuel/
change_biof_mandate.gms

```
*******************************************************************************
$ontext

   CAPRI project

   GAMS file : change_biof_mandate.gms

   @purpose  : increase/decrease biofuel mandates uniformly for all EU Member States

   @author   : Mihaly Himics
   @date     : 14.10.19
   @since    :
   @refDoc   :
   @seeAlso  :
   @calledBy :

$offtext
*******************************************************************************


*
*   ---    push upwards the sigmoid function for the biofuel share in total fuel demand
*          see equation biofDemShare_ for more details
*


*
*   ---    take over shock from the main scenario file
*
$setlocal mandate_shock %1


  parameter p_newMandate(rall, xxbiof) 'adjusted biofuel mandate in absolute terms';


  p_newMandate(RMS, xxbiof) $ ((RMS_TO_RM(RMS,'A_EU_WEST') or RMS_TO_RM(RMS,'A_EU_EAST'))
                               $ DATA(RMS,'QUTS',XXBIOF,'CUR'))
    = DATA(RMS,"QUTS",XXBIOF,"CUR") * %mandate_shock%;

  DATA(RMS,"QUTS",XXBIOF,'AbsoluteLevel') $ p_newMandate(RMS, xxbiof)
    = p_newMandate(RMS, xxbiof) ;


  option p_newMandate:2:1:1;
  display "New mandates/targets/quotas: ", p_newMandate;
```

3. Exercise: Run the scenario



3. Exercise: Analyze results in GUI



3. Exercise: Analyze results in GUI

- Check if scenario shock is correctly transmitted
- Find the right table



3. Exercise: Analyze results in GUI

- Check if scenario shock correctly transmitted



3. Exercise: Analyze results in GUI

a. Biofuel balances
Identify impacts on…

- Biodiesel production: ____%
- Biodiesel consumer price: __%
- Energy share: +/- ____ percentage points



	mandate_SA_150defaulta	
	Bio diesel	Bio ethanol
Total Biofuel production [1000 t]	32418.58 33.19%	12708.61 71.17%
First Generation Biofuels (from Agriculture) [1000 t]	14730.34 121.45%	8879.42 146.98%
Second Generation Biofuels [1000 t]	15677.95 0.00%	3829.10 0.00%
Biofuels from non-agricultural sources [1000 t]	2010.30 0.00%	0.10 0.00%
Biofuel-use by transport sector [1000 t]	47944.05 124.69%	16330.81 125.06%
Biofuel-use by industry [1000 t]		
Energy share of biofuels in total transportation fuel use [% of energy content]	30.43 117.77%	25.93 118.30%
Mandate or Quota obligation for biofuels share in total transportation fuel use [% of energy content]	30.40 141.74%	25.92 142.51%
Imports [1000 t]	15732.25 786.96%	3629.09 1685.56%
Exports [1000 t]	206.78 -95.67%	6.89 -98.14%
consumer prices [Euro/ton]	1982.27 130.18%	3697.10 230.98%
consumer taxes [Euro/Ton]	61.07 -0.77%	140.32 0.09%

3. Exercise: Analyze results in GUI

b. Biofuel feedstock



3. Exercise: Analyze results in GUI

b. Biofuel feedstock

- Which biodiesel feedstock use increased the most in relative terms: ___
- Which biodiesel feedstock use increased the least in relative terms: ___
- Possible impact on traded feedstock?

3. Exercise: Analyze results in GUI

c. Market balances with non-EU countries



3. Exercise: Analyze results in GUI

c. Market balances with non-EU countries

- Larger import/smaller export of biofuel feedstock
- Can you identify sustainability issues with EU biodiesel production?



	Net production [1000 t]	Human consumption plus losses [1000 t]	Processing [1000 t]	Biofuels processing [1000 t]	Feed use [1000 t]	Imports without intra trade [1000 t]	Exports without intra trade [1000 t]	Net trade [1000 t]
Rapeseed oil	13031.53 8.01%	2004.41 -3.47%	1825.07 -42.83%	8614.45 64.72%	117.65 -60.89%	76.11 65.91%	546.05 -58.36%	469.94 -62.87%
Sunflower seed oil	4451.21 6.71%	2713.52 -5.95%	894.40 -14.13%	808.06 122.95%	47.61 -57.14%	1717.97 225.28%	1705.59 469.63%	-12.38 94.59%
Soya oil	2117.88 -3.26%	1101.93 -0.33%	615.94 -6.31%	1373.82 202.86%	174.87 -43.23%	1401.72 63.27%	253.03 -51.63%	-1148.69 -242.49%
Olive oil	2431.58 0.45%	1900.91 2.65%	64.86 -0.52%			127.24 1.41%	642.47 2.13%	515.23 2.30%
Palm oil		438.02 0.47%	4962.64 -7.74%	6334.13 274.77%		11734.79 56.36%		-11734.79 -56.36%
Other oil	2932.22 0.00%	1170.65 2.99%	3790.36 -0.30%		2.32 10.04%	2631.37 -0.01%	600.27 -3.70%	-2031.10 -1.13%

Sources

- Becker, Arno (2011), Impacts of European biofuel policies on global biofuel and agricultural markets. Dissertation at Rheinische Friedrich-Wilhelms-Universität zu Bonn. https://bonndoc.ulb.uni-bonn.de/xmlui/handle/20.500.11811/4751?locale-attribute=en.
- Blanco, María, Adenäuer, Marcel, Shrestha, Shailesh and Becker, Arno (2013), Methodology to assess EU Biofuel Policies: The CAPRI Approach, JRC Scientific and Policy Reports. Scientific and Technical Research series –ISSN 1831-9424 (online). https://publications.jrc.ec.europa.eu/repository/handle/JRC80037.
- Yano, Yuki, Blandford, David and Surry, Yves R., (2012), From Ethanol Shuffle to Ethanol Tourism— Why the RFS Does Not Make Sense, Choices: The Magazine of Food, Farm, and Resource Issues, 27, issue 4, number 143189, https://EconPapers.repec.org/RePEc:ags:aaeach:143189.

Advanced biofuels

ANNEX IX

Part A. Feedstocks for the production of biogas for transport and advanced biofuels, the contribution of which towards the minimum shares referred to in the first and fourth subparagraphs of Article 25(1) may be considered to be twice their energy content:

(a) Algae if cultivated on land in ponds or photobioreactors;

(b) Biomass fraction of mixed municipal waste, but not separated household waste subject to recycling targets under point (a) of Article 11(2) of Directive 2008/98/EC;

(c) Biowaste as defined in point (4) of Article 3 of Directive 2008/98/EC from private households subject to separate collection as defined in point (11) of Article 3 of that Directive;

(d) Biomass fraction of industrial waste not fit for use in the food or feed chain, including material from retail and wholesale and the agro-food and fish and aquaculture industry, and excluding feedstocks listed in part B of this Annex;

(e) Straw;

(f) Animal manure and sewage sludge;

(g) Palm oil mill effluent and empty palm fruit bunches;

(h) Tall oil pitch;

(i) Crude glycerine;

(j) Bagasse;

(k) Grape marcs and wine lees;

(l) Nut shells;

(m) Husks;

(n) Cobs cleaned of kernels of corn;

(o) Biomass fraction of wastes and residues from forestry and forest-based industries, namely, bark, branches, pre-commercial thinnings, leaves, needles, tree tops, saw dust, cutter shavings, black liquor, brown liquor, fibre sludge, lignin and tall oil;

(p) Other non-food cellulosic material;

(q) Other ligno-cellulosic material except saw logs and veneer logs.

Part B. Feedstocks for the production of biofuels and biogas for transport, the contribution of which towards the minimum share established in the first subparagraph of Article 25(1) shall be limited and may be considered to be twice their energy content:

(a) Used cooking oil;

(b) Animal fats classified as categories 1 and 2 in accordance with Regulation (EC) No 1069/2009.